

RATSUB

A recursive graphics language

Adam Ponting

v1.0 October 2019

Tutorial and language specification

updated Dec 2024

Contents

1	Ratsub tutorial	4
	Getting started	4
	Levels	5
	The sketch	5
2	The structure of RATSUB programs	6
	Preamble	6
	Defining initial points	6
	Shape definitions	7
	Defining new points	7
	Defining multiple new points	8
	Shape calls	8
3	Colouring	9
	Basic colouring	9
	Basic shape colouring	9
	Basic shape call colouring	10
	Random colouring	10
	Random range colouring	10
	Shape shading	10
	Initial point shading	10
	Gradient shading	11
	Shape call colouring	12
	Colour-add mode	12
	Random	12
	Colour operators	12
4	Random numbers	13
	rand	13
	crand	13
	srand	13
	Random range	13
5	More commands	14
	draw	14
	grid	14
	Turning left and right	14
	flip	14
	Wait shapes and wait	15
	Conditionals	15
6	Drawing in 3D with pcube	17
7	RATSUB language reference	18
	Levels	18
	Program structure	18
	Program structure in detail	20
	Optional settings	20
	Initial point definitions	20

	Shape definitions	20
8	Colour names	23

1. Ratsub tutorial

Ratsub is a programming language for drawing pictures using recursion and subdivision. A Ratsub program describes an initial shape, how to subdivide it into other shapes, and how long to keep subdividing. The shapes are made of points, which are joined with straight lines. When a shape reaches the maximum recursion level, specified in the shell command, the edge or interior of the shape, or both, are drawn.

Getting started

For writing Ratsub programs you need three things—pen and paper to do sketches, a shell terminal to run commands, and a text editor to enter programs.

Open a text editor to enter Ratsub programs into, like SublimeText, VS Code, TextEdit, NotePad, TextWrangler etc—not a word processor, which may save text as rich text, MS Word or some other format. We want plain text files.

The Ratsub program on the right (yes, that’s the whole program!) produces a black square with a white border.

Save it as `test.sdv` in the `subdiv` folder. In the shell, `cd` to the `subdiv` folder, then running the shell command `ratsub test 0` should produce a PNG image in the same folder. In the terminal you should see something like this:

```
def sq 4
```

```
$ ratsub test 0
test 0
Making test-0.png...
Done.
```



If nothing happened, the error message should say why not, or at least where in the program the compiler thinks an error is.

Next, type in this program, pressing ENTER after every line, including the last:

The indentation after the first line isn’t necessary—it just makes the programs easier to read.


Save it as `test.sdv`, then on the command line enter:

```
def sq 4
  rand
  p4..7 p0.. 1/2 p1..0
  sq :415c
  sq :c526
  sq :04c7
  sq :7c63
```

```
$ ratsub test 0 4
```

Five `.png` images, named `test-0.png` to `test-4.png`, should appear in the `subdiv` folder.

The images should look something like this:



Levels

As a Ratsub program runs, shapes are always called with an implicit level parameter, initially 0 for the first shape in the program, and a number of point parameters, which are implicit for the first shape.

A Ratsub program must be called with a level parameter indicating the maximum number of recursion levels to perform. The first shape in the program is level 0 the first time. The shapes it calls (other shapes, or itself, or both) are level 1. The shapes they call are level 2, etc.

When a shape is called at the maximum recursion level, it is drawn with its current colour setting. Nothing is drawn until the maximum level is reached—except using the draw command, which draws a shape immediately.

The shape calls are done *depth-first*, i.e. the first shape called goes all the way to max recursion depth before the second shape is called. And the order shapes are called affects what is on top, what is underneath. That can change the appearance of some programs entirely.

After 10 or 20 levels, the shapes are usually too small too see, and can take a long while to draw if there are millions of them.

The sketch

The first step in writing a Ratsub program is to *make a sketch* of your idea on paper. Draw the initial shape and the shapes it will call or be divided into. Then number the points, preferably in a way that can be efficiently coded.

2. The structure of RATSUB programs

A Ratsub program is made of an optional preamble, followed by one or more shape definitions.

Initial points are defined with x,y coordinates, and other points are affine combinations of them, e.g.

`p4 p0 .5 p1`

The `.5` means `p4` will be the average or midpoint of the two, halfway from `p0` to `p1` — mathematically, $\frac{p0 + p1}{2}$. Changing the `.5` to various other numbers will get you points on the line between `p0` and `p1`.

`p4 p0 0 p1` means `p4` is at `p0`.

`p4 p0 1 p1` means `p4` is at `p1`.

`p4 p0 2 p1` means go from `p0` to `p1`, then go the same distance again to reach `p4`.

You need to use `left` and `right` (see p. 14) to make new points off the line between `p0` and `p1`.

Preamble

Defining initial points

Initial points became the parameters of the first shape, the first time through. A square needs 4 initial points, a hexagon 6 etc.

A single initial point is set like so:

`p0 100 100`

They must be numbered in order, starting from 0.

DEFAULTS: If `p0` is omitted, it's set to 0,0. If no initial points are set at all, they default to the 4 points of `pbox`, i.e. `pbox 0 0 100 100`.

`pbox`

The four points forming a (non-tilted) rectangle can be set with

`pbox a b c d`

where `a,b` is the lower left vertex and `c,d` the upper right.

Short form is short for

`pbox c d` `pbox 0 0 c d`

`pbox` `pbox 0 0 100 100`

pgon

`pgon n r θ`

This sets the vertices of a regular polygon as the initial points. e.g.

```
pgon 5 50 20
```

makes `p0` to `p4` the 5 vertices of a regular pentagon, centred at (0,0) with radius 50, with `p0` at 20° up from the x-axis, and the other points following anti-clockwise.

DEFAULTS: If θ is omitted, it's set to 0, i.e. `p0` will be at (r,0).

Shape definitions

Defining new points

New points can be defined by saying how far the point is along the line from one existing point to another, e.g.

```
p2 p0 1/2 p1
```

defines a new point `p2`, halfway from `p0` to `p1`. This also can be written

```
p2 p0 0.5 p1
```

or

```
p2 p0 .5 p1
```

par

```
p3 par p0 p1 p2
```

New points can also be defined using `par`, which here makes a new point `p3`, the same direction and distance from `p2` as `p1` is from `p0`. i.e. it draws the 4th point of a parallelogram, given the first 3.

New point numbering

A shape's first new point must be numbered to come immediately after the shape's parameter points (or the initial points, for the first shape). e.g. for

```
def square 4
```

the parameter points—the points the shape is called with—will be named `p0`, `p1`, `p2` and `p3`, and the first new point must be `p4`.

Defining multiple new points

Short form	Equivalent to
p4..6 p0.. .5 p1..	p4 p0 .5 p1 p5 p1 .5 p2 p6 p2 .5 p3
p4..6 p0.. .5 p1..0	p4 p0 .5 p1 p5 p1 .5 p2 p6 p2 .5 p0

That second form is handy for creating midpoints on each side of a shape with one command.

Shape calls

shape points options

A shape may have no shape calls. That means *draw it right away*, without waiting for the level to reach the max.

Short forms

Short form	Short for
<i>shape</i> p1..5	<i>shape</i> p1,2,3,4,5
<i>shape</i> :14253	<i>shape</i> p1,4,2,5,3

N.B. The :14253-type short form can only be used if every point in the list has a single-digit number! p8,9,10 cannot be abbreviated to :8910.

3. Colouring

Colours are represented as *RGB* (red green blue), with the 3 values between 0 (off) and 1 (fully on) inclusive. Colours can be set with R G B values or with a colour name.

Colour can be applied in Ratsub in two different places in a shape definition:

- shape colour - usually in the line below the initial `def` line.

```
def sq 4
  red
  tri :012
```

- shape call colour. This sets the called shape's colour.

```
def sq 4
  tri :012 red
```

Unless `onlyedges` is set, every shape is drawn with some colour. A shape's colour is either

1. its shape colour, if it has one
2. the colour given to it by the shape call it was called with,
3. otherwise, the default colour—either black, or mid-grey (0.5,0.5,0.5) if colour-add mode is used.

Colouring syntax examples:

METHOD	SHAPE COLOURING	SHAPE CALL COLOURING
basic	red	tri :012 red
initial point shading	ip0 red	
gradient shading	p0 red p1 green	
colour-add mode		tri :012 + 1 2 -1
colour operator		tri :012 not

Basic colouring

Basic shape colouring

r g b
c (same as *c c c*)
colourname

DEFAULT: Black.

Basic shape call colouring

If a shape call has a colour, e.g.

```
sq :0123 red
```

it will pass that colour, otherwise it will pass its shape's colour, if it has one. Otherwise it will pass the colour its shape was passed when it was called.

Random colouring

rand

crand

rand makes a random colour, different each time through the shape/shape call. crand is the same random colour each time.

Random range colouring

```
rand  $r_{min}$   $r_{max}$   $g_{min}$   $g_{max}$   $b_{min}$   $b_{max}$ 
```

```
crand  $r_{min}$   $r_{max}$   $g_{min}$   $g_{max}$   $b_{min}$   $b_{max}$ 
```

These make a random colour with red in the range $r_{min} \dots r_{max}$, etc.

Shape shading

Apart from giving each shape a fixed flat colour, there are two ways to colour shapes with a varying shade.

Initial point shading

```
ipn  $r$   $g$   $b$ 
```

```
ipn colourname
```

Initial point shading colours shapes by how close they are to a set of initial points.

Specify, in a shape's definition, a set of initial points (the fixed points defined in the preamble) and their colours. When the shape is called, it's coloured by these points, proportional to how close it is to them.

Extra initial points to control shading can be added to the preamble, and they will be ignored by the first shape, which only takes as many points as it's defined to need.


Multiple shading colours can be defined on one line.

Level 3

```

def s1 4
  ip0 red ip1 blue
  ip2 dark green ip3 black
  p4..7 p0.. .5 p1..0
  s2 :415c
  s1 :c526
  s1 :04c7
  s2 :7c63
def s2 4
  ip0 cyan ip1 yellow
  ip2 violet ip3 white
  p4..7 p0.. .5 p1..0
  s2 :415c
  s1 :c526
  s1 :04c7
  s2 :7c63

```



Gradient shading


Similar to initial point shading, but uses the points from the current shape.

Level 2

```

def sq 4
  p4..7 p0.. .5 p1..0
  sq :415c
  sq :c526
  sq :04c7
  sq :7c63
  =1
  wh :c415
  bl :c526
  bl :c704
  wh :c637
def bl 4
  p2 black p0 grey
def wh 4
  p0 white p2 grey

```



Suggestion: If you need to define new points to use with gradient shading, you can define them in the parent shape and include them as points in the shape call parameters.

Shape call colouring

Colour-add mode

shapename points + r g b

shapename points - r g b

shapename points + c

shapename points - c

This colours a shape depending on its call path down through the levels—on the add-colours of its ancestor shapes. It's used at the end of shape calls, e.g.

`sq :0123 + 1 2 3`

makes the red of the new `sq` shape 1/10th brighter, the green 2/10ths brighter, and the blue 3/10ths brighter. If the present shape's colour is (R,G,B), this calls `sq` with the colour (R+r/10,G+g/10,B+g/10). Because the colour adjustments are usually small fractions, the units are 10ths.

`sq :0123 + 1` means the same as `sq :0123 + 1 1 1`.

N.B. The initial shape takes its shape colour, but after that any shape colours are ignored in favour of the colour passed to it by the shape that called it.

Random

shapename points rand

Colour operators

These are used in shape calls to transform the shape's colour

Operator	
<code>not/!</code>	$R\ G\ B \Rightarrow 1-R\ 1-G\ 1-B$. Black (0 0 1) becomes white (1 1 1), blue (0 0 1) becomes yellow (1 1 0) etc.
<code>rot</code>	Rotate colours to the right. $R\ G\ B \Rightarrow B\ R\ G$.
<code>rot2</code>	Rotate right twice/rotate left. $R\ G\ B \Rightarrow G\ B\ R$.

4. Random numbers

Random numbers can be used in shape definitions for setting colours, and with distances, for setting new points. *Random ranges* allow fine control over the desired random numbers.

rand

`p6 p0 rand p1`

— this places `p6` randomly somewhere on the line between `p0` and `p1` (including perhaps on those points). This random number is different every time the shape is called.

crand

`p6 p0 crand p1`

— this places `p6` randomly somewhere on the line between `p0` and `p1` (including perhaps on those points). This random number is constant - the same every time the shape is called.

srand

`srand n`

— this seeds the random number generator. With the same `n`, the random numbers (and thus image) will be the same each time the program's run. For random numbers different each time without having to use `srand`, use `random` and `crandom` instead of `rand`.

Random range

new points

colours

5. More commands

draw

There are two ways of drawing a shape immediately—regardless of current level:

- Write `draw` instead of a shape name in a shape call. If a `draw` call has a colour, e.g.

`draw :0123 red`

it will use that colour, otherwise it will use its shape's colour, if it has one. Otherwise it will use the colour its shape was passed when it was called.

- A shape without any shape calls will be drawn immediately.

`Draw` is most useful when the child shapes only cover part of the parent, or lie outside it.

grid

`grid θ`

grid transforms the usual square grid into a grid with the y-axis θ degrees from the x-axis. It only affects the initial points. e.g.

Turning left and right

`newpointdef left a`

`newpointdef right a`

Turning is the only way to make a new point that's not on the line between its 2 defining points. `p2 p0 .5 p1 left .25`

This travels halfway from `p0` to `p1`, then turns left and travels half that distance—the `.25` means 1/4 the distance *between* `p0` and `p1`.

flip

`shapecall flip`

Used to call a flipped-over version of a shape, especially a line, for example in line fractals like the Koch curve, and IFS.

Flipping a shape only makes a difference if left or right are used in the shape's new point definitions.

Wait shapes and wait

Used at the end of shape calls:

`wait` makes a wait shape, i.e. wait 1 level then call.

`wait n` makes n linked wait shapes, i.e. wait n levels then call.

temp

Give example of pics needing 2, 3 wait shapes. Also multiple wait shapes, calls to different numbers in different places.

Conditionals

`>n` `=n` `>-n` `=-n`

With `>` or `=`, you can do different shape calls depending on the current level.

```
def sq 4
  tri :012
  tri :032
  >2
  tri :123
  tri :103
```

This does the first two shape calls if the current level ≤ 2 , and the second two if it's >2 . i.e. :

If level >2 then do C and D, else do A and B

A negative branch number $-n$ means n levels before the maximum level.

```
def sq 4
  tri :012
  tri :032
  >-2
  tri :123
  tri :103
```

Here the second branch is taken when the current level is more than 2 less than the maximum level. For example, if the program is run with "ratsub myprog 8" then the second branch is taken when the current level >6 .


Similarly, `=5` means "take this second branch when the current level is 5"; `=-5` means "take this second branch when the current level is 5 less than the maximum level".

```

width .02
def sq 4
  p4 c
  p5..8 p0.. .5 p1..0
  sq :5164
  sq :4627
  sq :0548
  sq :8473
  =-2
  sq1 :0123
  draw :0123 green
def sq1 4
  p4..7 p0.. rand p1..0
  draw :4567 rand

```

Level 7




```

margin 6
def frame 4
  p4 c
  p5..8 p4 1.1 p0..
  sq :0123
  backgr :5678
def backgr 4
  p0 yellow p2 blue
def sq 4
  p4 c
  p5..8 p0.. .5 p1..0
  sq :5164
  sq :4627
  sq :0548
  sq :8473
  =-2
  sq1 :0123
def sq1 4
  p4..7 p0.. rand p1..0
  shard :4567
def shard 4
  p4 p3 .5 p0
  p5 p1 .5 p2
  p4 blue p5 yellow

```

Level 7



6. Drawing in 3D with pcube

`pcube r θ_x θ_y θ_z $z_{\text{viewpoint}}$ $z_{\text{viewplane}}$`

This adds 8 initial points, the vertices of a cube seen in perspective. The image is drawn as if the viewer's eye is at $(0,0,z_{\text{viewpoint}})$ and the viewplane is the plane $z = z_{\text{viewplane}}$.

N.B. The algorithm involves dividing by $z - z_{\text{viewpoint}}$, so will make an error if that's 0, i.e. if you get so close that your eye hits a vertex of the cube!


7. RATSUB language reference

Levels

Each Ratsub program specifies a (theoretically infinite) number of pictures, depending on the level parameter which sets the maximum recursion level/call stack depth. The first shape in a program calls others, they call others etc. When the recursion reaches N levels deep, a shape instead draws itself using its parameter points and colour settings.

This can be bypassed with the draw command in a shape's definition—the shape will be drawn immediately.

Program structure



Structure of a SHAPE DEFINITION	1. def <i>shape</i> <i>n</i>		
	2. optional	$\left[\begin{array}{l} \text{colour setting} \left[\begin{array}{l} \text{colourname} \\ r \ g \ b \\ c \end{array} \right] \\ \text{shading points} \left\{ \begin{array}{l} \text{ipn colourname} \\ \text{ipn } r \ g \ b \end{array} \right\} \end{array} \right.$	
	3. 0 or more new point definitions	$\left\{ \begin{array}{l} \text{single} \left\{ \begin{array}{l} \text{pn } c \\ \text{pn } pq \ a/b \ pr \ [\text{TURN}] \\ \text{pn } pq \ a \ pr \ [\text{TURN}] \end{array} \right\} \\ \text{multiple} \left\{ \begin{array}{l} \text{pm} \dots n \ pq[\dots] \ a \ pr[\dots][s] \ [\text{TURN}] \\ \text{pm} \dots n \ pq[\dots] \ a \dots b/c \ pr[\dots][s] \ [\text{TURN}] \end{array} \right\} \end{array} \right.$	
	4. 0 or more SHAPE CALLS		
	5. optional	$\left\{ \begin{array}{l} \text{branching} \left[\begin{array}{l} >n \\ >-n \\ =n \\ =-n \end{array} \right] \\ \text{then 0 or more SHAPE CALLS} \end{array} \right.$	

TURN $\left[\begin{array}{l} \text{left } a \\ \text{right } a \end{array} \right.$

A SHAPE CALL $\left\{ \begin{array}{l} \left[\begin{array}{l} \text{shapename} \\ \text{draw} \end{array} \right. \\ \text{points} \left\{ \begin{array}{l} \text{pn}_0, n_1, \dots, n_k \\ :n_0 n_1 \dots n_k \end{array} \right\} \\ \text{optional} \left\{ \begin{array}{l} \text{colour} \left[\begin{array}{l} \text{colourname} \\ r \ g \ b \\ \text{not} \\ \text{rot} \\ \text{rot2} \\ = \\ +/- \ r \ g \ b \\ +/- \ c \end{array} \right] \\ \left[\begin{array}{l} \text{wait} \\ \text{wait } n \end{array} \right] \\ \text{flip} \end{array} \right\} \end{array} \right.$

Program structure in detail

Optional settings

<code>grid θ</code>	Makes the angle between the x and y axes θ degrees, for the initial point definitions.
<code>margin <i>num</i></code>	set width of the margin around the picture
<code>margin <i>l t r b</i></code>	set width of the left, top, right, bottom margin
<code>margin <i>a b</i></code>	<i>a</i> is left/right margin, <i>b</i> is top/bottom margin.
<code>no edges</code>	no shape edges are drawn
<code>only edges</code>	only edges are drawn
<code>background <i>r g b</i></code>	colour of the background bounding box
<code>width <i>num</i></code>	set line width for drawing shape edges.

Initial point definitions

These define the vertices of the initial shape, which are initially passed as parameters to the first shape definition. If `grid` is set then these points will be affected. Use " to repeat an x- or y-coordinate from the line above.

<code>p3 5.5 2.3</code>	defines point 3 at 5.5,2.3
<code>pbox <i>a b c d</i></code>	defines four points, the corners of a rectangle numbered clockwise from lower left. The new points <code>p0 ... 3</code> are: <i>a,b a,d c,d c,b</i> .
<code>pbox <i>c d</i></code>	short for <code>pbox 0 0 <i>c d</i></code>
<code>pbox</code>	short for <code>pbox 0 0 100 100</code>
<code>pgon <i>N r θ</i></code>	Defines the <i>N</i> points of a polygon, with one more point at the centre 0,0. The point have radius <i>r</i> and <code>p0</code> is at θ degrees. The <i>N</i> points are numbered anticlockwise.
<code>pcube <i>r $\theta_x \theta_y \theta_z$ zviewpoint zviewplane</i></code>	Defines the 8 points of a cube in 3D.

Shape definitions

These define shapes and how they'll be subdivided into other shapes. Shapes can also be subdivided into themselves, as in the example program above.

Initial point definitions and the first shape

Ratsub manual: The first shape defined is drawn with the initial points from the preamble, using as many points as it needs, e.g. "def sq 4" defines a shape "sq" with 4 points, which will

take p_0 , p_1 , p_2 and p_3 of the initial points the first time it's called, and the 4 points it's called with each time after that. So only the first shape defined can take initial points as parameters. After the first shape has been called once, the initial points can only be used by shapes for their shading. "p0" then means, the first parameter in the call to that shape, and not the first initial point.

The def line

`def shape N`

N is the number of points in the shape, the number parameters it takes from the shape that calls it. The first shape, the first time it's run when the program starts, takes the initial points (defined in the preamble) as its parameters.

Shape colour setting (optional)

`colourname` set the fill colour.

`r g b` set the shape colour to red r , green g , blue b . (colour values are between 0 and 1)

`c` set the shape colour to red c , green c , blue c .

`rand` random colour

`+ Δr Δg Δb`

`− Δr Δg Δb` use colour-add mode

Shading point definitions

`ip0 r g b` Assigns the colour $r g b$ to the initial point 0, the first defined in the preamble (Not the first parameter of the shape!)

`ip0..3 r g b` Assign the same colour to initial points 0,1,2,3.

New point definitions

	Defines...
<code>p2 p0 1/3 p1</code>	a new point p_2 , 1/3 of the way from p_0 to p_1 .
<code>p2 p0 0.4 p1</code>	a new point p_2 , 4/10 of the way from p_0 to p_1 .
<code>p2 p0 rand crand random p1</code>	a new random point somewhere on the line between p_0 and p_1 .

New point definition short forms

p3 p1 " p2 uses the same fraction/number as the previous line.

p4..6 p0 1..3/4 p1	p4 p0 1/4 p1
	p5 p0 2/4 p1
	p6 p0 3/4 p1

p4..6 p0 1,2,4/5 p1	p4 p0 1/5 p1
	p5 p0 2/5 p1
	p6 p0 4/5 p1

p5..7 p1.. .95 p0	p5 p1 .95 p0
	p6 p2 .95 p0
	p7 p3 .95 p0

p5..7 p0 .95 p1..	p5 p0 .95 p1
	p6 p0 .95 p2
	p7 p0 .95 p3

p5..7 p0.. .95 p1..	p5 p0 .95 p1
	p6 p1 .95 p2
	p7 p2 .95 p3

p5..7 p0.. .95 p1..0	p5 p0 .95 p1
	p6 p1 .95 p2
	p7 p2 .95 p0

This last form is frequently useful, e.g. for the midpoints of the sides of a pentagon :

<pre> pgon 5 50 0 def pent 5 p5..9 p0.. 1/2 p1..0 </pre>
--

as the last side is from p4 back to the first point p0.

8. Colour names

These colour names can be used in Ratsub instead of RGB colour values.

alice blue	dim grey	medium aquamarine
antique white	dodger blue	medium blue
aqua	firebrick	medium orchid
aquamarine	floral white	medium purple
azure	forest green	medium sea green
beige	gainsboro	medium slate blue
bisque	ghost white	medium spring green
black	gold	medium turquoise
blanched almond	golden rod	medium violet red
blue violet	gray	midnight blue
blue	green yellow	mint cream
brown	green	misty rose
burly wood	grey	moccasin
cadet blue	honeydew	navajo white
chartreuse	hot pink	navy
chocolate	indian red	old lace
coral	indigo	olive drab
corn silk	ivory	olive
cornflower blue	khaki	orange red
crimson	lavender blush	orange
cyan	lavender	orchid
dark blue	lawn green	pale golden rod
dark cyan	lemon chiffon	pale green
dark golden rod	light blue	pale turquoise
dark gray	light coral	pale violet red
dark green	light cyan	papaya whip
dark grey	light golden rod yellow	peach puff
dark khaki	light gray	peru
dark magenta	light green	pink
dark olive green	light grey	plum
dark orange	light pink	powder blue
dark orchid	light salmon	purple
dark red	light sea green	red
dark salmon	light sky blue	rosy brown
dark sea green	light slate gray	royal blue
dark slate blue	light slate grey	saddle brown
dark slate gray	light steel blue	salmon
dark slate grey	light yellow	sandy brown
dark turquoise	lime green	sea green
dark violet	lime	sea shell
deep pink	linen	sienna
deep sky blue	magenta	silver
dim gray	maroon	sky blue

slate blue
slate gray
slate grey
snow
spring green
steel blue

tan
teal
thistle
tomato
turquoise
violet

wheat
white smoke
white
yellow green
yellow